

Decomposing interactions using GLM in combination with the COMPARE, LMATRIX and MMATRIX subcommands in SPSS

Glen T. Howell
Carleton University

Guy L. Lacroix
Carleton University

In this tutorial, we provide researchers who use SPSS step-by-step instructions for decomposing interactions when a three-way ANOVA is conducted using the GLM procedure. We start with a demonstration of how a two-way interaction can be decomposed using the COMPARE subcommand in combination with syntax. Then, we provide instructions with examples for conducting simple interaction and second-order simple effects analyses for three-way ANOVAs with between-subjects, within-subjects, and mixed between- and within-subjects variables using the LMATRIX or MMATRIX subcommands. Provided in Appendices are general rules that can be used to derive design-specific LMATRIX and MMATRIX subcommands.

Since the release of SPSS 7.5, the General Linear Model (GLM) procedure has become the only way of conducting analyses of variance (ANOVAs) in SPSS without resorting to syntax (Nichols, 1997). Introduced in replacement of the MANOVA procedure, GLM allows users to conduct between-subjects, within-subjects, and mixed-design ANOVAs by following the steps prescribed in a short series of dialog boxes that are fairly intuitive to use. Moreover, in addition to generating *F* Tables, GLM lets users run a wide range of complementary tests and statistics. Hence, by clicking relevantly labeled options, users can obtain tests of assumptions, descriptive statistics, power analyses, effect

sizes, and plots. Finally, GLM allows users to perform post hoc analyses when significant main effects are found. Once again, these tests are easy to access via the dialog boxes, they are explicitly labeled, and in the case of between-subjects designs, the number of available tests is large (e.g. LSD, Scheffe, Tukey...). Hence, it is not surprising that many introductory textbooks now describe how to use GLM and how to interpret its outputs (e.g., Field, 2005; Tabachnick & Fidell, 2007)¹.

Unfortunately, the GLM procedure in SPSS does not offer a simple, intuitive, and user-friendly way of decomposing interactions when factorial designs are used. This is especially true when the ANOVA includes three independent variables. Hence, the goal of this tutorial paper

Glen T. Howell and Guy L. Lacroix, Department of Psychology, Carleton University, Ottawa, Canada.

The authors would like to thank Tyler Burleigh and Nicole Robert for reviewing a previous version of this paper. This research was made possible by grants awarded to the second author by the Natural Sciences and Engineering Research Council of Canada (NSERC).

Correspondence concerning this tutorial may be addressed to Glen Howell, Department of Psychology, Carleton University, B550 Loeb Building, 1125 Colonel By Drive, Ottawa, ON, Canada, K1S 5B6.

¹ Nonetheless, GLM may still not be the most popular way to run ANOVAs in SPSS. An internet search conducted November 17th, 2010 using Google generated 106 000 hits for the keywords "MANOVA SPSS", whereas it generated 85 500 hits for the keywords "GLM SPSS". MANOVA probably remains popular because it preceded GLM and it works well. Hence, for researchers who do not mind using syntax to run all their analyses, there is no incentive to change procedures.




Figure 1. Decision Tree for conducting post hoc analyses following a two- or three-way ANOVA.

is to provide step-by-step instructions explaining how to decompose three-way ANOVAs using the GLM procedure in combination with the COMPARE, LMATRIX, and MMATRIX subcommands. The most recent version of SPSS (known as PASW Statistics 18) will be used throughout this tutorial. We will proceed as follows.

We will begin with a short introduction to the GLM procedure. We will then present a decision tree that helps the reader choose which follow-up analyses to use when a three-way ANOVA is conducted, depending on the significance of the different interactions and main effects. We will start with a demonstration of how to analyze simple effects using the COMPARE subcommand when a significant two-way interaction is found. At this point, we will introduce the Syntax Editor. We will then describe how the LMATRIX and MMATRIX subcommands can be used in combination with GLM to decompose a significant three-way interaction. To illustrate the procedure, we will present three step-by-step examples, including one that involves only between-subjects variables, one that involves one between- and two within-subjects variables, and one that involves only within-subjects variables.

Before moving on, we offer the following advisory notes. First, this tutorial assumes that readers are familiar with complex ANOVA designs. Consequently, we are voluntarily brief in presenting theory. Second, we assume that readers are familiar with SPSS and that, in most cases, will have previously used GLM. Third, the reader should remember that it is impossible to provide examples for all factorials designs. Thus, it is probable that readers will not find an

example which exactly replicates the research design that is of concern to them in this tutorial. Hopefully, however, it will be possible for them to customize the syntax presented herein to meet their analytical needs. Finally, this tutorial focuses on decomposing interactions in SPSS, and therefore does not demonstrate how to conduct post hoc pairwise comparisons following a significant main effect (Many references including Field, 2005 describe how SPSS may be used to conduct post hoc analyses).

How does GLM work?

For the purpose of this paper, we will not describe the theory behind the GLM, nor the algorithms used. We will, however, provide a brief description of how the GLM works in order to help the reader understand how the LMATRIX and MMATRIX can be customized to test specific contrasts. Most generally, contrasts on a user specified GLM test whether the null hypothesis for a linear combination of parameter estimates is likely to be true. For a between-subjects design, the null hypothesis is $H_0: LB = K$, where L represents the contrast coefficients matrix (LMATRIX), B represents a vector of estimated parameters, and K represents the contrast results matrix (KMATRIX). By default, the LMATRIX is the estimated function for the intercept and the KMATRIX is 0 (SPSS inc., 2009a). As we will illustrate later on, a researcher may customize the LMATRIX to test any contrast on the parameter estimates by specifying a corresponding set of weighting coefficients in the Syntax Editor. As such, it is possible to test the null hypothesis that the user-specified linear combination of




Figure 2. Two-way interaction between one within- and one between-subjects variable.

weighted parameter estimates is 0.

For a within-subjects or mixed-design, the null hypothesis is $H_0: LBM = K$, where the new term M represents the transformation coefficients matrix (MMATRIX). By default, the MMATRIX corresponds to the average transformation matrix for the dependent variables (i.e., the levels of the within-subjects variables; SPSS inc., 2009b). As we will show later on, a researcher may customize LMATRIX or MMATRIX in a within-subject or mixed-design, to test any contrast or comparison of interest by specifying a set of weighting coefficients for these matrices in the Syntax Editor. By customizing LMATRIX or MMATRIX, the GLM procedure tests the null hypothesis that the specified linear combination of weighted parameter estimates is 0.

Decision tree

Throughout this tutorial, we will use the decision tree presented in Figure 1 in combination with plots of means of fictional data to show how to decompose significant two- and three-way interaction effects. The decision tree is meant to be a guideline for conducting post hoc analyses on a two- or three-way factorial design. To facilitate comprehension, we will work backwards through the decision tree, starting with a significant two-way interaction.

The reader should note that this tutorial illustrates a simple effects approach to the decomposition of interaction effects. This is not the only way to decompose interaction effects, however. For example, Keppel (1991, Chapters 12 & 20) describes an interaction comparisons and contrasts approach. Although this latter approach is not adopted in this tutorial, the LMATRIX and MMATRIX subcommands in GLM can be used to conduct it.

A two-way interaction is found

In this next section, we will explain how to analyze simple effects when a two-way interaction is found. We will also introduce the Syntax Editor. This example will involve the data found in the file case1.sav.

The appropriate design to analyze these data is a $2 \times 2 \times (3)$ ($A[1, 2] \times B[1, 2] \times C[1, 2, 3]$) mixed-design ANOVA, with C as a within-subjects variable². Under the "Analyze" menu, the researcher would go to "General Linear Model" and then select "Repeated Measures...". Next, in the "Repeated Measures Define Factor(s)" dialog box, the researcher would provide a name for the within-subjects variable in the "Within-Subject Factor Name:" box, put the number of levels in the "Number of Levels:" box, and click on "Add" then "Define". Then, in the "Repeated Measures dialog box", the researcher would put the levels of the within-subjects variable in the "Within-Subjects Variables (name):" box and the between-subjects variables in the "Between-Subjects Factor(s):" box, then click "OK". In the output, the tables of Tests of Within-Subjects Effects and Tests of Between-Subjects Effects show that the three-way interaction is not significant ($p = .89$) and only the two-way $B \times C$ interaction is significant ($p < .001$). A plot of the two-way interaction shown in Figure 2 was generated using the "Plots..." menu in the "Repeated Measures" dialog box.

The reader can infer from the plot that the interaction stems from the difference between the first and second levels of B at the second and third levels of C . To verify this impression, the decision tree presented in Figure 1 suggests that the researcher should analyze first-order simple effects. To do so, the researcher would have two choices: examine the impact of C on each of the levels of B or examine the impact of B on each of the three levels of C . In an actual research situation, the choice between these two options would be guided by theoretical considerations. Throughout this tutorial, however, the choice that directly tests the impression formed when examining the plot of means will be made. Hence, for the present example, we will examine the simple effect of B at each of the three levels of C .

To examine these simple effects, the researcher would first define the mixed-design ANOVA and then click the "Options..." button in the "Repeated Measures" dialog window. The next step is to send the $B \times C$ interaction into the "Display Means for:" box and click "Continue". The researcher would then click the "Paste" button instead of the

² For clear and concise instructions for formatting data files for repeated-measures analyses in SPSS, the reader may refer to Lacroix and Giguère (2006).




Figure 3. Syntax Editor with Compare added to run simple effects tests of B on each level of C.

“OK” button in the “Repeated Measures” dialog window. This opens a Syntax Editor window instead of running the analysis (see Figure 3). The B*C interaction that was sent

Univariate Tests

Measure:MEASURE_1

C		Sum of Squares	df	Mean Square	F	Sig.	Partial Eta Squared	Noncent. Parameter	Observed Power ^a
1	Contrast	3.366E-29	1	3.366E-29	.000	1.000	.000	.000	.050
	Error	.240	20	.012					
2	Contrast	150.000	1	150.000	961.538	.000	.980	961.538	1.000
	Error	3.120	20	.156					
3	Contrast	599.000	1	599.000	3792.702	.000	.995	3792.702	1.000
	Error	3.159	20	.158					

Each F tests the simple effects of B within each level combination of the other effects shown. These tests are based on the linearly independent pairwise comparisons among the estimated marginal means.

a. Computed using alpha = .05

Figure 4. Univariate Tests table displaying the simple effects of B at each level of C.

into the “Display Means for:” box generated an “/EMMEANS = TABLES(B*C)” line in the syntax. This line of syntax creates descriptive statistics for the interaction. It also may be customized to generate simple effects tests. Specifically, the researcher would type “COMPARE” one space after the “/EMMEANS = TABLES(B*C)” line and define the to-be-decomposed variable within parentheses. In Figure 3, “COMPARE(B)” was added in order to test the simple effect of B at each level of C. To run the ANOVA and all subcommands, the researcher simply needs to highlight the syntax with the cursor and run it by right clicking and then selecting the “Run Current” option. Researchers may use this procedure for a simple effects analysis on a two-way interaction for a between-subjects, within-subjects, or mixed-design.

When the analysis is run, the output includes a

Univariate Tests table (see Figure 4) and a Pairwise Comparisons table because a between-subjects variable was chosen³. The output shows that the difference between the two levels of B is not significant at the first level of C ($p = 1.00$), but is significant at the second and third levels of C (both $ps < .001$). Because B only has two levels, the Pairwise Comparisons table is redundant⁴. These results confirm the impression formed when examining Figure 2. Specifically, the results show that the two-way B × C interaction stems from the fact that B has a significant impact on the second and third levels of C, but not at the first level of C.

The reader may wonder why we did not simply rerun the ANOVA using the Univariate procedure in GLM with B in the “Fixed Factor:” box and hold the level of C constant by putting C1, C2, or C3 in the “Dependent Variable:” box. This approach would provide the Mean Square Error for the selected simple effect, whereas the COMPARE approach provides the Mean Square Error for the overall A × B × C interaction effect, which is considered more stable because it involves all cells in the factorial design instead of only those cells involved in the selected simple effect. However, if there are large discrepancies among the error variances for the

different levels of C (noted by a violation of homogeneity of variance), the reader would be well advised to rerun three separate ANOVAs on each level of C instead of using the

³ If the within-subjects variable were chosen (i.e., C), the output would include a Multivariate Tests table instead of a Univariate Tests table, with one test for each of the two simple effects of C on B.

⁴ By default, the LSD method for multiple comparisons is used. If the researcher wishes to control for alpha inflation as a result of performing more tests than there are degrees of freedom for the effect, the Bonferroni or Sidak adjustment may be used. To do so, the researcher needs to type “ADJ” with either BONFERRONI or SIDAK enclosed in brackets one space after the COMPARE subcommand (e.g., ADJ(BONFERRONI)).




Figure 5. Three-way interaction among between-subjects variables.

COMPARE method (Keppel, 1991; Tabachnick & Fidell, 2007). Although COMPARE can be easily used in combination with syntax to compute simple effects, as we have demonstrated, it cannot be used to decompose higher-order interaction effects, unlike the LMATRIX and MMATRIX subcommands which will now be introduced.

A three-way interaction is found

In this next section, we will examine three examples that feature a significant three-way interaction. Each, in turn, will be decomposed using the LMATRIX or MMATRIX subcommands in GLM. The following three ANOVA designs will be analyzed in order: three-way between-subjects, three-way mixed-design, and three-way within-subjects.

Three-way between-subjects ANOVA

The data used for the first example can be found in case2.sav. A $3 \times 2 \times 2$ ($A[1, 2, 3] \times B[1, 2] \times C[1, 2]$) between-subjects ANOVA is appropriate to analyze the data. After defining and running the ANOVA in "Univariate...", the researcher would see that the Tests of Between-Subjects Effects table in the output shows that the three-way interaction is significant ($p < .001$). Figure 5 displays a plot of the means. Following the decision tree presented in Figure 1, the next step is to test simple interaction effects. The reader may infer from the plot that the three-way interaction stems from the difference between the two levels of B at the second and third levels of A, but only at the first level of C. To confirm this impression, we will first examine the simple $A \times B$ interaction on each of the two levels of C. The procedure is similar for the two other possible sets of simple interaction

effects (i.e., $B \times C$ on each of the levels of A and $A \times C$ on each of the levels of B).

As was done when the two-way interaction is found example was examined, the researcher must first paste the syntax that runs the analysis and go to the Syntax Editor. Then, the LMATRIX subcommand may be typed in after the last line of syntax (one must first erase the period, then add it at the end of the LMATRIX statement; see Figure 6). By default, the LMATRIX, or contrast coefficient matrix, used in computation of effects in GLM is equal to the estimable function for the intercept matrix. Changing the contrast

Figure 6 is a screenshot of the PASW Statistics Syntax Editor window. The title bar reads '3wayBetween.sps - PASW Statistics Syntax Editor'. The menu bar includes 'File', 'Edit', 'View', 'Data Transfer', 'Analysis', 'Direct', 'Mark', 'Graph', 'Utilities', 'Add-on', 'Run', 'Tools', 'Window', and 'Help'. The toolbar contains various icons for file operations and analysis. The main text area contains the following syntax:

```

3  GLM DV BY A B C
4  /METHOD=SSTYPE(3)
5  /INTERCEPT=INCLUDE
6  /EMMEANS=TABLES(A*B*C)
7  /CRITERIA=ALPHA(.05)
8  /DESIGN=A B C A*B A*C B*C A*B*C
9  /LMATRIX "simple interaction A x B at C1"
10 A*B  -1 -1  -1  1  0  0
11 A*B*C  1  0  -1  0  -1  0  0  0  0;
12 A*B  -1 -1  0  0  -1  1
13 A*B*C  1  0  -1  0  0  0  0  0  -1  0
14 /LMATRIX "simple interaction A x B at C2"
15 A*B  -1 -1  -1  1  0  0
16 A*B*C  0  1  0 -1  0 -1  0  1  0  0;
17 A*B  -1 -1  0  0  -1  1
18 A*B*C  0  1  0 -1  0  0  0  0  -1  0.

```

The status bar at the bottom indicates 'PASW Statistics Processor is ready' and 'ln 24 Col 41'.

Figure 6. LMATRIX syntax for conducting simple interaction effects on between-subjects variables.

coefficient matrix using the LMATRIX subcommand allows the researcher to make customized contrasts, including simple interaction effects, second-order simple effects, and second-order simple comparisons. The LMATRIX and MMATRIX subcommands require the researchers to identify specific effects by providing weighting coefficients. Appendix A describes a general procedure for deriving the weighting coefficients required for simple interaction and second-order simple effects analyses. The general rules for creating the correct LMATRIX syntax for these analyses that can be generalized to any three-way between-subjects design are described in Appendix B. For the upcoming examples, however, only the most relevant aspects of the particular contrasts involved in decomposing the significant three-way interactions will be discussed.

After the LMATRIX subcommand has been typed in the Syntax Editor, the researcher must define the contrast with weighting coefficients for each of the effects involved in the computation of the contrast. Figure 6 displays the syntax for the overall analysis with the LMATRIX subcommands that test the simple $A \times B$ interaction at C1 and at C2. As can be seen in the two LMATRIX subcommands, there is a brief description of each contrast on lines 9 and 14 contained in quotation marks (single quotation marks may also be used to write programmer notes to be ignored by the syntax editor). These descriptions are optional, but important to include, because they clearly identify the contrasts in the output. Below the description for the first LMATRIX subcommand on line 9 are two linearly independent contrasts with the first one defined on lines 10 and 11 and the second one on lines 12 and 13. These contrasts are separated by a semi-colon at the end of line 11, which tells SPSS that two contrasts need to be tested. In this case, two contrasts are needed to use the two degrees of freedom for the simple $A \times B$ interaction effect (see Appendix B for further details). Each contrast needs to be defined within the $A \times B$ effect as well as the higher-order $A \times B \times C$ effect.

For the first contrast, the six digits following the A^*B effect on line 10 represent the weighting coefficients that will be applied to the six cell means corresponding to the three levels of A crossed with the two levels of B. The reader must note that the weighting coefficients are ordered in a way that is consistent with how the GLM is defined on line 3. Specifically, the reader may see that A is presented before B after the "BY" statement where the independent variables are listed. Because the variables are presented this way, the levels of A change more slowly in the left to right order of the cells than the levels of B. In the present example, the six digits represent the coefficients for, from left to right, the A1B1, A1B2, A2B1, A2B2, A3B1, and A3B2 cells. The reader may see that all levels of the faster changing variable (B) are

paired with the first level of the slower changing variable (A), before moving on to the second and third levels of A. Hence, if B was presented before A on line 3, the six digits would represent the coefficients for the A1B1, A2B1, A3B1, A1B2, A2B2, and A3B2 cells. For defining any contrast using the LMATRIX subcommand, this rule concerning the order of the cells must be respected to ensure accurate calculation of the contrast. All the cells that are not involved in the contrast have a 0 value (i.e., the A3B1 and A3B2 cells), while those that are involved have either a 1 or a -1⁵. After examining the weighting coefficients for the A^*B effect, the reader may see that the contrast compares the effect of B on the first level of A (i.e., A1B1 - A1B2) with the effect of B on the second level of A (i.e., A2B2 - A2B1).

Next, the researcher must also define the A^*B effect in the weighting coefficients for the A^*B^*C effect, but only at the first level of C. Like the A^*B effect, the weighting coefficients need to be ordered in a way that is consistent with how the GLM is defined. Specifically, the crossing of the three levels of A, with the two levels of B, and then two levels of C, yields 12 cells in the following abbreviated order from left to right with A changing slowest and C changing fastest, A1B1C1, A1B1C2, A1B2C1, A1B2C2, A2B1C1..., A3B2C2. To fix the contrast at C1, the weighting coefficients for all cells that include C2 are assigned a 0. To finalize the set of coefficients, the cells that correspond to the previously defined A^*B effect are assigned the respective weighting coefficients only for cells that include C1 (i.e., A1B1C1, A1B2C1, A2B1C1, and A2B2C1).

The second contrast defined on lines 12 and 13 compares the effect of B at the first level of A with the effect of B at the third level of A. The weighting coefficients for the A^*B and A^*B^*C effects for this contrast change slightly from the preceding ones. The reader may notice that the change simply represents the switch from comparing the effect of B at A1 with the effect of B at A3, instead of at A2. The two linearly independent contrasts encompass the simple interaction effect $A \times B$ at C1, which we will soon see is reflected in the output when we discuss the second LMATRIX subcommand. The second LMATRIX subcommand defined on lines 14 to 18 tests the simple interaction $A \times B$ at C2. The A^*B effects in the two contrasts defined on lines 15 and 17 are identical to the ones in the first LMATRIX subcommand. However, the reader may notice that the weighting coefficients for the A^*B^*C effects

⁵ LMATRIX accepts fractions, numbers greater than 1, and numbers less than -1. However, in the simple interaction effects and second-order simple effects presented in this paper, only 0, 1 and -1 are used.

Custom Hypothesis Tests #1

Contrast Results (K Matrix)^a

Contrast		Depe...	
		DV	
L1	Contrast Estimate	-9.926	
	Hypothesized Value	0	
	Difference (Estimate - Hypothesized)	-9.926	
	Std. Error	.194	
	Sig.	.000	
	95% Confidence Interval for Difference	Lower Bound	-10.314
		Upper Bound	-9.537
L2	Contrast Estimate	-9.925	
	Hypothesized Value	0	
	Difference (Estimate - Hypothesized)	-9.925	
	Std. Error	.194	
	Sig.	.000	
	95% Confidence Interval for Difference	Lower Bound	-10.314
		Upper Bound	-9.537

a. Based on the user-specified contrast coefficients (L) matrix: simple interaction A x B at C1

Test Results

Dependent Variable: DV

Source	Sum of Squares	df	Mean Square	F	Sig.
Contrast	197.031	2	98.516	1741.900	.000
Error	3.393	60	.057		

Figure 7. Tests results for the simple A x B interaction at C1.

on lines 16 and 18 have changed, simply to reflect that C has been fixed at level 2 instead of level 1.

The output corresponding to the simple A x B interaction at C1 is displayed in Figure 7. This section of the output is labeled Custom Hypothesis Tests #1, reflecting that these are the results for the first user-defined LMATRIX. As a secondary reminder, just beneath the Contrast Results (KMATRIX) table appears a note that the results in the table refer to the user-defined LMATRIX labeled "simple interaction A x B at C1". The Contrast Results table displays the contrast estimate, standard error, p-value, and 95% confidence interval for the first (L1) and second (L2) contrasts. For our purpose, these results are not of interest.

The Test Results table below, however, displays the results for a simultaneous test of both contrasts, which is the simple A x B interaction at C1. The results show that this effect is significant ($p < .001$). However, the Test Results table in the Custom Hypothesis Tests #2 section, which has been excluded for brevity, shows that the simple A x B interaction at C2 is not significant ($p = .97$).

Following the decision tree (see Figure 1), the next step is to decompose the significant simple A x B interaction at C1 using second-order simple effects analyses. There are two different avenues of approach here: Either one can test the second-order simple effects of A at each level of B and C1 or the second-order simple effects of B at each level of A and

```

3wayBetween.sps - PASW Statistics Syntax Editor
File Ed Vie Dat Transfr Analy Direct Mar Grap Utiliti Add-c Ru Too Wind Hel
[*] [Red Circle] [Green Checkmark] [Speaker] [Printer] [Save] [Open] [Close] [Home] [Back] [Forward] [Refresh] [Download] [Upload] [Data] [Tools] [Help]
Active: DataSet9
29 GLM DV by A B C
30 /METHOD=SSTYPE(3)
31 /INTERCEPT=INCLUDE
32 /EMMEANS=TABLES(A*B*C)
33 /CRITERIA=ALPHA(.05)
34 /DESIGN=A B C A*B A*C B*C A*B*C
35 /LMATRIX "second-order simple effect of B at A1 and C1"
36   B 1 -1
37   A*B 1 -1 0 0 0 0
38   B*C 1 0 -1 0
39   A*B*C 1 0 -1 0 0 0 0 0 0 0
40 /LMATRIX "second-order simple effect of B at A2 and C1"
41   B 1 -1
42   A*B 0 0 1 -1 0 0
43   B*C 1 0 -1 0
44   A*B*C 0 0 0 0 1 0 -1 0 0 0 0
45 /LMATRIX "second-order simple effect of B at A3 and C1"
46   B 1 -1
47   A*B 0 0 0 0 1 -1
48   B*C 1 0 -1 0
49   A*B*C 0 0 0 0 0 0 0 0 1 0 -1 0.
PASW Statistics Processor is ready In 51 Col 27 NUM

```

Figure 8. LMATRIX syntax for conducting second-order simple effects on between-subjects variables.

C1. In order to confirm the impression formed when examining Figure 5, the second-order simple effects of B at each level of A and C1 will be computed. The LMATRIX subcommands required to run these analyses are presented in Figure 8.

Three LMATRIX subcommands are required to test the second-order simple effects of B at each of the three levels of A and C1. Within each LMATRIX subcommand, there is only one contrast because there is only one degree of freedom for B. Let us examine the first LMATRIX on lines 35 to 39. The B effect is listed first, which is the primary comparison of interest. The weighting coefficients are set to compare the first level of B with the second level of B. The researcher may note that the B effect has also been defined in the context of the higher-order interactions involving B (i.e., the A*B, B*C, and A*B*C effects, in order). Recall that the variable listed earliest in the line of syntax defining the

GLM changes slower in order of the cells than variables listed later. Hence, the order of the weighting coefficients for the A*B and the B*C effects need to respect the following order: A1B1, A1B2, A2B1, A2B2, A3B1, and A3B2 as well as B1C1, B1C2, B2C1, and B2C2, respectively. For the A*B effect, the reader may verify that the coefficients reflect a comparison of B1 with B2 at the first level of A. Similarly, for the B*C effect, the reader may verify that the coefficients reflect a comparison of B1 with B2 at the first level of C. Lastly, for the A*B*C effect, the comparison of the levels of B on A1 at C1 is represented in the weighting coefficients by the assignment of the value 1 for A1B1C1, -1 for A1B2C1, and 0 for the ten other coefficients. The next two LMATRIX subcommands define the second-order simple effects of B at A2 and C1 (see lines 40 to 44) as well as B on A3 at C1 (see lines 45 to 49) respectively, using the same procedure described above.




Figure 9. Three-way interaction among one between- and two within-subjects variables.

When the syntax is run, the output includes three Custom Hypothesis Tests sections, each with a Contrast Results and Test Results table. The Test Results tables show that the first contrast, the second-order simple effect of B at A1 and C1, is not significant ($p = 1.00$), while the other two second-order simple effects are significant (both $ps < .001$). Hence, as was suspected when examining the plot in Figure 5, the three-way interaction stems from the difference between the two levels of B at the second and third levels of A, but only at the first level of C.

Three-way mixed-design ANOVA

In the second example, we show how to decompose a three-way interaction when the design has both between- and within-subjects variables. In this case, the MMATRIX subcommand must be used in conjunction with the LMATRIX subcommand. Specifically, an MMATRIX subcommand defines the contrast in relation to the within-subjects variable(s), while, as we have already seen, an LMATRIX subcommand defines the contrast in relation to the between-subjects variable(s). The general rules for creating the correct LMATRIX and MMATRIX for these analyses for any three-way mixed-design are described in Appendix C. Before showing how the MMATRIX subcommand functions, let us analyze the data in case3.sav using a $2 \times (2) \times (2)$ ($A[1, 2] \times B[1, 2] \times C[1, 2]$) mixed-design ANOVA with B and C as within-subjects variables. First, the researcher may proceed to define the repeated measures ANOVA in GLM and click the "Paste" button instead of the "OK" button in the "Repeated Measures" dialog box. Then, the syntax should be selected and executed.

In the output, the results in the Tests of Within-Subjects Effects table show that the three-way $A \times B \times C$ interaction is significant ($p < .001$). Figure 9 presents a plot of the means. The pattern of means suggests that the three-way interaction stems from the difference between the two levels of A at the second level of B and second level of C. To verify this observation, we will work through the decision tree presented in Figure 1 and first test the simple $C \times A$ interaction at each of the levels of B. Figure 10 displays the pasted syntax with the LMATRIX and MMATRIX subcommands required to test these simple interaction effects.

The reader may first notice on lines 12 and 14 that each contrast in the MMATRIX subcommand has a label defined using quotation marks. The syntax that follows the labels on lines 13 and 15 defines the within-subjects portion of the simple interaction effects. The "all" statement tells SPSS that all combinations of the levels of the within-subjects variables at that line of syntax will be assigned weighting coefficients. Specifically, for the present example, the order of the cells from left to right is B1C1, B1C2, B2C1, and B2C2. The reader may confirm that the first contrast on line 13 defines the effect of C on B1⁶ (i.e., B1C1 - B1C2). Now, because the

⁶ The researcher may also list the names of the combined levels of the within-subjects variables (as defined in the dependent variables portion of the GLM line of syntax) in any order with corresponding weighting coefficients. For example, after the label, write "A1B2 -1 A1B1 1" and get the same outcome. Names not listed are assigned a 0 by default.

```

3  GLM B1C1 B1C2 B2C1 B2C2 BY A
4  /WSFACTOR=B 2 Polynomial C 2 Polynomial
5  /METHOD=SSTYPE(3)
6  /PLOT=PROFILE(C*A*B)
7  /CRITERIA=ALPHA(.05)
8  /WSDESIGN=B C B*C
9  /DESIGN=A
10 /LMATRIX "A1 vs. A2"
11      A 1 -1
12 /MMATRIX "simple interaction C x A at B1"
13      all 1 -1 0 0;
14      "simple interaction C x A at B2"
15      all 0 0 1 -1.

```

Figure 10. LMATRIX and MMATRIX syntax for conducting simple interaction effects on a combination of between- and within-subjects variables.

LMATRIX subcommand requests a comparison between A1 and A2, the contrast tested for the first MMATRIX

subcommand is the combined comparison of A1 and A2 with C1 and C2 at B1 (i.e., the simple $C \times A$ interaction at B1). The second contrast defined in the MMATRIX subcommand on line 15 requests the effect of C on B2 (i.e., $B2C1 - B2C2$). Hence, combined with the LMATRIX subcommand, this line of syntax tests the simple $C \times A$ interaction at B2. The pertinent results are displayed in Figure 11.

In the Custom Hypothesis Tests #1 section, the reader may focus on the Contrast Results (K Matrix) and Univariate Test Results tables, while ignoring the Multivariate Test Results table that has been omitted from Figure 11. In the Contrast Results (K Matrix) table, we find two contrasts under the Transformed Variable heading: the first one shows that the simple $C \times A$ interaction at B1 is not significant ($p = 1.00$) while the second shows that the simple $C \times A$ interaction at B2 is significant ($p < .001$). The corresponding F -tests are presented in the Univariate Tests Results table.

Following the decision tree illustrated in Figure 1, we will proceed by testing second-order simple effects. Based on our observation from examining the plots in Figure 9, we will proceed to test the second-order simple effect of A at C1 and B2 as well as A at C2 and B2. The syntax for running these analyses is presented in Figure 12. The reader may see that there is one LMATRIX subcommand on line 35 and one MMATRIX subcommand on line 37. The LMATRIX

Custom Hypothesis Tests #1

Contrast Results (K Matrix)^a

Contrast	Transformed Variable	
	simple interaction C x A at B1	simple interaction C x A at B2
L1 Contrast Estimate	.000	10.000
Hypothesized Value	0	0
Difference (Estimate - Hypothesized)	.000	10.000
Std. Error	.424	.762
Sig.	1.000	.000
95% Confidence Interval for Difference	Lower Bound	- .891
	Upper Bound	.891

a. Based on the user-specified contrast coefficients (L) matrix: A1 vs. A2

Univariate Test Results

Source	Transformed Variable	Sum of Squares	df	Mean Square	F	Sig.
Contrast	simple interaction C x A at B1	.000	1	.000	.000	1.000
	simple interaction C x A at B2	500.000	1	500.000	172.414	.000
Error	simple interaction C x A at B1	16.200	18	.900		
	simple interaction C x A at B2	52.200	18	2.900		

Figure 11. Tests results for the simple $C \times A$ interaction on each of the levels of B.

```

28 GLM B1C1 B1C2 B2C1 B2C2 BY A
29 /WSFACTOR=B 2 Polynomial C 2 Polynomial
30 /METHOD=SSTYPE(3)
31 /PLOT=PROFILE(C*A*B)
32 /CRITERIA=ALPHA(.05)
33 /WSDESIGN=B C B*C
34 /DESIGN=A
35 /LMATRIX "A1 vs. A2"
36     A 1 -1;
37 /MMATRIX "second-order simple effect A at C1 and B2"
38     all 0 0 1 0;
39     "second-order simple effect A at C2 and B2"
40     all 0 0 0 1.

```

Figure 12. LMATRIX and MMATRIX syntax for conducting second-order simple effects of a between-subjects variable on within-subjects variables.

subcommand requests a comparison of the first level of A with the second level of A. The first contrast in the MMATRIX subcommand on line 38 fixes the comparison of A at B2C1 while the second contrast on line 39 fixes the

comparison at B2C2. With the combined LMATRIX and MMATRIX subcommands, we obtain the particular second-order simple effects of interest.

The output presented in Figure 13 contains a Custom Hypothesis Tests #1 section. Within the section, the reader will find a Contrasts Results (K Matrix) and Test Results table displaying the outcome for both second-order simple effects tests. Both tables show that the second-order simple effect of A at C1 and B2 is not significant ($p = 1.00$) while the second-order simple effect of A at C2 and B2 is significant ($p < .001$). Hence, the decomposition of the three-way interaction using simple effects analyses confirms our impression that the interaction stems from the difference between the two levels of A at the second level of B and second level of C.

Three-way within-subjects ANOVA

For the third example, we decompose a significant three-way interaction using MMATRIX when the design is fully within-subjects. The general rules for creating the correct MMATRIX for these analyses for any three-way within-subjects design are described in Appendix D. The data for the present example are located in case4.sav. The appropriate design to analyze them is a $(3) \times (2) \times (2)$ ($A[1, 2, 3] \times B[1, 2] \times C[1, 2]$) within-subjects ANOVA. The researcher may proceed to define the repeated measures

Custom Hypothesis Tests #1

Contrast Results (K Matrix)^a

Contrast		Transformed Variable	
		second-order simple effect A at C1 and B2	second-order simple effect A at C2 and B2
L1	Contrast Estimate	.000	-10.000
	Hypothesized Value	0	0
	Difference (Estimate - Hypothesized)	.000	-10.000
	Std. Error	.424	.365
	Sig.	1.000	.000
	95% Confidence Interval for Difference		
	Lower Bound	-.891	-10.767
	Upper Bound	.891	-9.233

a. Based on the user-specified contrast coefficients (L) matrix: A1 vs. A2

Univariate Test Results

Source	Transformed Variable	Sum of Squares	df	Mean Square	F	Sig.
Contrast	second-order simple effect A at C1 and B2	.000	1	.000	.000	1.000
	second-order simple effect A at C2 and B2	500.000	1	500.000	750.000	.000
Error	second-order simple effect A at C1 and B2	16.200	18	.900		
	second-order simple effect A at C2 and B2	12.000	18	.667		

Figure 13. Test results for second-order simple effects of a between-subjects variable on within-subjects variables.




Figure 14. Three-way interaction among three within-subjects variables.

ANOVA in GLM, click the “Paste” button instead of the “OK” button, then select and run the syntax.

The results show that the three-way interaction is significant ($p < .001$). Figure 14 presents a plot of the means. The reader may infer from the plots that the three-way interaction stems from the difference between the two levels of C at the second level of B and the first level of A. Following the decision tree (see Figure 1), we should proceed with tests of simple interaction effects. Based on our impression of the nature of the three-way interaction, we will test the simple $B \times C$ interactions on each of the three levels of A.

The syntax required to run this analysis is presented in Figure 15. The reader may notice that there is a single

MMATRIX subcommand with three labeled and specified contrasts on lines 10 to 15. The contrasts are defined with the weighting coefficients required to test, in order: the simple $B \times C$ interaction at the first, second, and then third level of A (lines 11, 13, and 15 respectively). The weighting coefficients have been assigned in the same manner as previously described. The results for these analyses are presented in Figure 16. In the Custom Hypothesis Tests section, the reader may find the Contrast Results (K Matrix) and Univariate Test Results tables, which contain the results for each simple interaction effect. The results show that the simple $B \times C$ interaction at A1 is significant ($p < .001$), while the simple $B \times C$ interactions at A2 and at A3 are not significant ($p = 1.00$ and $p = .43$, respectively).




Figure 15. MMATRIX syntax for conducting simple interaction effects on within-subjects variables.

To further decompose the three-way interaction and second-order simple effect of C at B1 and A1 as well as C at confirm our impression, the researcher would test the B2 and A1. The syntax to run these analyses is presented in

Custom Hypothesis Tests

Contrast Results (K Matrix)

Contrast ^a	Transformed Variable		
	simple interaction B x C at A1	simple interaction B x C at A2	simple interaction B x C at A3
L1 Contrast Estimate	17.000	.000	.300
Hypothesized Value	0	0	0
Difference (Estimate - Hypothesized)	17.000	.000	.300
Std. Error	.533	.533	.371
Sig.	.000	1.000	.428
95% Confidence Interval for Difference	Lower Bound	15.884	-1.116
	Upper Bound	18.116	1.116

a. Estimable Function for Intercept

Univariate Test Results

Source	Transformed Variable	Sum of Squares	df	Mean Square	F	Sig.
Contrast	simple interaction B x C at A1	5780.000	1	5780.000	1016.852	.000
	simple interaction B x C at A2	.000	1	.000	.000	1.000
	simple interaction B x C at A3	1.800	1	1.800	.655	.428
Error	simple interaction B x C at A1	108.000	19	5.684		
	simple interaction B x C at A2	108.000	19	5.684		
	simple interaction B x C at A3	52.200	19	2.747		

Figure 16. Tests results for the simple B x C interaction on each of the three levels of A.

```

*3wayWithin.sps - PASW Statistics Syntax Editor
File Edit View Data Transform Analyze Direct Marketing Graphs Utilities Add-ons Run Tools Window Help
Active: DataSet5
26 GLM A1B1C1 A1B1C2 A1B2C1 A1B2C2 A2B1C1 A2B1C2 A2B2C1 A2B2C2 A3B1C1 A3B1C2 A3B2C1 A3B2C2
27 /WSFACTOR=A 3 B 2 C 2
28 /METHOD=SSTYPE(3)
29 /EMMEANS=TABLES(A*B*C)
30 /CRITERIA=ALPHA(.05)
31 /WSDESIGN=A B C A*B A*C B*C A*B*C
32 /MMATRIX "second-order simple effect C at B1 and A1"
33     all 1 -1 0 0 0 0 0 0 0 0 0 0;
34     "second-order simple effect C at B2 and A1"
35     all 0 0 1 -1 0 0 0 0 0 0 0 0.
PASW Statistics Processor is ready In 37 Col 26

```

Figure 17. MMATRIX syntax for conducting second-order simple effects on within-subjects variables.

Figure 17. There is one MMATRIX subcommand on line 32 with two specified contrasts on lines 33 and 35. The first contrast has weighting coefficients that request the second-order simple effect of C at B1 and A1 and the second contrast requests the second-order simple effect of C at B2 and A1. The output shows that the second-order simple effect of C at B1 and A1 is not significant ($p = .16$) while the second-order simple effect of C at B2 and A1 is significant ($p < .001$). Thus, the three-way interaction, as we suspected from examining the plots in Figure 14, stems from the difference between the levels of C at the second level of B and first level of A.

Summary

In this tutorial, we have provided researchers who use the GLM procedure in SPSS explicit instructions for decomposing two- and three-way interactions. Specifically, we showed how to generate the proper COMPARE, LMATRIX and MMATRIX subcommands to conduct simple interaction and first- and second-order simple effects analyses for three different types of ANOVA designs. The additional instructions and syntax provided in the Appendices offer general rules for creating customized design-specific syntax. With the syntax, instructions, and illustrative examples provided, researchers who prefer to use GLM in SPSS may continue to use it for conducting ANOVAs when significant two- or three-way interactions are found. Alternatively, researchers could learn the MANOVA procedure in SPSS, which does not offer the point and click method, and write all the syntax required for analysis or use another perhaps less intuitive data analysis

program.

If the reader desires to practice conducting the simple interaction and second-order simple effects analyses presented in this tutorial, the data files and corresponding syntax are available on the Tutorials for Quantitative Methods for Psychology website at <http://www.tqmp.org>. The reader may find the SPSS data and syntax files by choosing the content menu and then clicking on the word "sample" that follows the title of this paper.

References

- Field, A. (2005). *Discovering statistics using SPSS* (2nd Ed.). London: Sage.
- Keppel, G. (1991). *Design and analysis: A researcher's handbook, 3rd ed.* Englewood, Cliffs, NJ: Prentice-Hall.
- Lacroix, G. L., & Giguère, G. (2006). Formatting data files for repeated-measures analyses in SPSS: Using the Aggregate and Restructure Procedures. *Tutorials in Quantitative Methods for Psychology, Vol. 2*, 20-26.
- Maxwell, S. E., & Delaney, H. D. (2004). *Designing experiments and analysing data: A model comparison perspective* (2nd ed.). Mahwah, NJ: Erlbaum.
- Nichols, David P. (1997). From MANOVA to GLM: Basics of Parameterization. Retrieved August 6, 2008, from UCLA: Academic Technology Services, Statistical Consulting Group: <http://www.ats.ucla.edu/stat/Spss/library/manglm.htm>.
- Rodgers, J. L., Nicewander, W. A., & Toothaker, L. E. (1994). Linearly independent, orthogonal, and uncorrelated variables. *The American Statistician, 38*, 133-134.
- Tabachnick, B. G., & Fidell, L. S. (2007). *Experimental designs*

using ANOVA. Belmont, California: Thomson Brooks/Cole.

Toothaker, L. E. (1993). *Multiple comparison procedures.* Newbury Park, CA: Sage Publications. Series: Quantitative Applications in the Social Sciences, No. 89.

SPSS inc. (2009a). *PASW Statistics Base 18.* Chicago, IL: SPSS inc.

SPSS inc. (2009b). *PASW Advanced Statistics 18.* Chicago, IL: SPSS inc.

Manuscript received 17 March 2011;

Manuscript accepted 17 May 2011.

Appendices follow.

Appendix A.

Generating the weighting coefficients necessary for a set of linearly independent contrasts is paramount to creating the correct LMATRIX and MMATRIX. This task can prove to be difficult when simple interactions and second-order simple effects have more than two degrees of freedom. Guidelines are presented here, however, that may be generalized to simple interactions and second-order simple effects with many degrees of freedom. To illustrate the procedure, a $3 \times 3 \times 3$ ($A[1,2,3] \times B[1,2,3] \times C[1,2,3]$) design will be referenced throughout this section.

The researcher may first choose one of the variables involved in the effect of interest (e.g. a simple $A \times B$ interaction effect) and identify a set of pairwise comparisons on that variable. For the present illustration, we will use A and the pairwise comparisons commonly conducted when following-up on a significant main effect in ANOVA. The following comparisons, expressed as a list of weighting coefficients, are linearly independent and they together use the two degrees of freedom for A: $L1_A = 1 -1 0$ and $L2_A = 1 0 -1$ with the first, second, and third digits representing the coefficients for the A1, A2, and A3 cells respectively. These pairwise comparisons are linearly independent because neither one can be expressed as a linear combination of the other one. If the last logically possible pairwise comparison ($L3_A = 0 1 -1$) were included in the set, however, the set of three comparisons would not be linearly independent, because any one of them could be expressed as a linear combination of the other two (e.g., $L3_A = L2_A - L1_A$; see p. 178 in Maxwell & Delaney, 2004). Thus, for a variable with three levels such as A, the researcher may simply choose two of the three possible pairwise comparisons and be confident that the set is linearly independent. Generally, if A has x

number of levels, the researcher may choose the comparisons involving the first level of A with each of the other levels of A (e.g., when A has $x = 4$ levels: $L1 = 1 -1 0 0$, $L2 = 1 0 -1 0$, and $L3 = 1 0 0 -1$).

Second, for simple interaction effects, one needs to consider comparisons involving the levels of two variables simultaneously, not just one variable. When introducing another variable into the set of contrasts, say B, the researcher must find a set of linearly independent contrasts for that variable alone. The general rule provided above may be used. In this case, because B has three levels, we can use $L1_B = 1 -1 0$ and $L2_B = 1 0 -1$, the same ones used above for A.

Third, now that a set of linearly independent comparisons for both A and B have been identified, the researcher may derive the weighting coefficients for the four linearly independent contrasts for the simple interaction $A \times B$ using the following method. The researcher may draw four separate two-way cross classification tables, one for each linearly independent contrast, with as many rows as there are levels for A and as many columns as there are levels for B. The row variable in each table must be a variable that was listed before the column variable on the line of syntax that defines the GLM. For the present example, A was defined before B, so the row variable for the tables is A and the column variable is B. Figure A1 presents an example of how the tables could be constructed.

With three levels for A and three for B, each table is a 3×3 matrix with nine cells ordered from A1B1 in the top left cell to A3B3 in the bottom right cell. Beside the row header A are the weighting coefficients that correspond to either $L1_A$ (the top two tables) or $L2_A$ (the bottom two tables) and below the column header B are the weighting coefficients

		B					B		
		B1	B2	B3			B1	B2	B3
		1	-1	0			1	0	-1
A	A1	1	-1	0	1	1	0	-1	
	A2	-1	1	0	-1	-1	0	1	
	A3	0	0	0	0	0	0	0	
		1	-1	0			1	0	-1
A	A1	1	-1	0	1	1	0	-1	
	A2	0	0	0	0	0	0	0	
	A3	-1	1	0	-1	-1	0	1	

Figure A1. Weighting coefficients for a set of linearly independent contrasts for a 3×3 simple interaction.

that correspond to $L1_B$ (the two tables on the left) or $L2_B$ (the two tables on the right). The reader may see that the top left table is comprised of the $L1_A$ and $L1_B$ contrasts, while the other tables comprise the three other possible permutations of the two contrasts on A and the two contrasts on B. It is important that the researcher ensures that all possible permutations of the contrasts derived on the two independent variables are represented in the collection of tables. Most generally, if the researcher has an $A \times B$ simple interaction, the $(a - 1)$ linearly independent contrasts on A would each need to be paired with each of the $(b - 1)$ linearly independent contrasts on B, for a total of $(a - 1) \times (b - 1)$ tables.

Appendix B.

Testing simple interaction effects in a three-way between-subjects ANOVA using LMATRIX

For any three-way between-subjects design, the number of LMATRIX subcommands required to test simple interaction effects is equal to the number of levels for the variable that is being held constant. Suppose a researcher has a $3 \times 3 \times 3$ ($A[1,2,3] \times B[1,2,3] \times C[1,2,3]$) design and wishes to test the simple $A \times B$ interaction effects at each level of C. In this case, the researcher would specify three LMATRIX subcommands, one for each level of C, the variable that is being held constant. The number of contrasts that need to be defined within each LMATRIX is equivalent to the number of degrees of freedom for the simple interaction. Following our example, the researcher would need to specify four contrasts (i.e., $df_{AB} = df_A \times df_B = (a - 1) \times (b - 1) = (3 - 1) \times (3 - 1) = 4$). Note that the contrasts do not need to be orthogonal, although they can be, but must be linearly independent (see Rodgers, Nicewander, & Toothaker, 1984 for a discussion of linear independence and orthogonality). Appendix A provides a detailed description of how to derive the weighting coefficients for a set of linearly independent contrasts.

For any three-way between-subjects design, the weighting coefficients for each linearly independent contrast will be used to define two effects. The first effect describes the simple interaction (e.g., A^*B), while the second effect describes the simple interaction in the higher-order three-way interaction (e.g., A^*B^*C). Figure B1 presents the LMATRIX syntax required to test the simple interaction $A \times B$ at C1. The reader may note that the four contrasts presented in Figure A1 are defined using weighting coefficients, are separated by a semi-colon, and each contrast has an A^*B and A^*B^*C effect (see lines 10 to 15 for the first

The weighting coefficients in the tables are calculated by taking the cross product of each cell's row and column weighting coefficients. For example, in the top left table in Figure A1, the weighting coefficient for cell A1B1 is the product of the weighting coefficient that corresponds to A1 on the left and the weighting coefficient that corresponds to B1 above (i.e., $1 \times 1 = 1$). The four sets of weighting coefficients contained in the tables are linearly independent and can be used in conjunction with the LMATRIX subcommand to test any 3×3 simple interaction, including the simple interaction $A \times B$ from the illustrative example.

contrast). The reader may also see that the weighting coefficients for the A^*B effects are organized in a matrix format, which corresponds exactly to how the coefficients are presented in the tables in Figure A1. The A^*B^*C effects are organized similarly, except C is an additional column variable which could be presented under the B variable in Figure A1. In Figure B1, this yields the following order for the columns from left to right: B1C1, B1C2, B1C3, B2C1, ..., B2C3. The reader may verify that the first contrast corresponds to the contrast represented in the top left table in Figure A1, while the next three contrasts correspond to the top right (lines 16 to 21), bottom left (lines 22 to 27), and bottom right tables (lines 28 to 33). The LMATRIX syntax for the two other simple interaction effects ($A \times B$ at C2 and $A \times B$ at C3) have the exact same structure, except the A^*B^*C weighting coefficients are fixed at C2 and C3 instead of at C1.

Testing second-order simple effects in a three-way between-subjects ANOVA using LMATRIX

Generally, the number of LMATRIX subcommands needed to test second-order simple effects is equal to the number of levels where the effects are tested. If the LMATRIX syntax in Figure B1 were run, the output would show that the simple interaction $A \times B$ at C1 is significant ($p < .001$) while the simple interactions at C2 and C3 are not ($ps > .99$). Hence, the researcher would want to test the second-order simple effects of B at the different levels of A and C1 (or B at the different levels of A and C1). In this case, three LMATRIX subcommands are required, one for each of the three levels of A where the second-order simple effects are tested. The number of contrasts required within each LMATRIX is equivalent to the number of degrees of freedom for the second-order effect. Following the example,

```

9  /LMATRIX "simple interaction A x B at C1"
10 A*B 1 -1 0
11     -1 1 0
12     0 0 0
13 A*B*C 1 0 0 -1 0 0 0 0 0
14     -1 0 0 1 0 0 0 0 0
15     0 0 0 0 0 0 0 0 0;
16 A*B 1 0 -1
17     -1 0 1
18     0 0 0
19 A*B*C 1 0 0 0 0 0 -1 0 0
20     -1 0 0 0 0 0 1 0 0
21     0 0 0 0 0 0 0 0 0;
22 A*B 1 -1 0
23     0 0 0
24     -1 1 0
25 A*B*C 1 0 0 -1 0 0 0 0 0
26     0 0 0 0 0 0 0 0 0
27     -1 0 0 1 0 0 0 0 0;
28 A*B 1 0 -1
29     0 0 0
30     -1 0 1
31 A*B*C 1 0 0 0 0 0 -1 0 0
32     0 0 0 0 0 0 0 0 0
33     -1 0 0 0 0 0 1 0 0
34 /LMATRIX "simple interaction A x B at C2"

```

Figure B1. LMATRIX for a 3×3 simple $A \times B$ interaction at C1.

we would need two linearly independent contrasts on B, one for each degree of freedom ($df_B = (b - 1) = (3 - 1) = 2$). The two contrasts on B defined earlier in Appendix A may be used (i.e., $L1_B = 1 -1 0$ and $L2_B = 1 0 -1$). Once the B effect is defined in the LMATRIX subcommand, all higher-order interactions involving B must also be defined (i.e., $A*B$, $B*C$, and $A*B*C$).

Figure B2 displays the LMATRIX syntax for testing the second-order simple effect of B at A1 and C1. On line 68, the $L1_B$ contrast is defined and on lines 69 to 77 it is defined in the higher-order $A*B$, $B*C$, and $A*B*C$ effects. Likewise, on line 78, the reader may see that the $L2_B$ contrast is defined and on lines 79 to 87 it is also defined in the higher-order effects. To test the other two second-order simple effects, the LMATRIX syntax from lines 67 to 87 may be copied and pasted twice in a row starting on line 88. Then some minor modifications may be made to the second and third LMATRIX subcommands, in order to fix the second-order effects at A2 and A3 respectively. First, the researcher would want to change the label in quotation marks for second and third LMATRIX subcommands on lines 88 and 109 such that it corresponds to the second-order effect at A2 and C1 and at

```

67 /LMATRIX "second-order simple effect of B at A1 and C1"
68 B 1 -1 0
69 A*B 1 -1 0
70     0 0 0
71     0 0 0
72 B*C 1 0 0
73     -1 0 0
74     0 0 0
75 A*B*C 1 0 0 -1 0 0 0 0 0
76     0 0 0 0 0 0 0 0 0
77     0 0 0 0 0 0 0 0 0;
78 B 1 0 -1
79 A*B 1 0 -1
80     0 0 0
81     0 0 0
82 B*C 1 0 0
83     0 0 0
84     -1 0 0
85 A*B*C 1 0 0 0 0 0 -1 0 0
86     0 0 0 0 0 0 0 0 0
87     0 0 0 0 0 0 0 0 0
88 /LMATRIX "second-order simple effect of B at A2 and C1"

```

Figure B2. LMATRIX syntax for testing the second-order simple effect of B at A1 and C1.

A3 and C1 respectively. Second, to fix the second-order effect at A2, the researcher would simply need to swap the lines in the second LMATRIX that correspond to lines 69 and 70, in order to fix the contrast at A2. Likewise, for the $A*B*C$ effect, the researcher would need to swap the lines that correspond to lines 75 and 76. Note that the $B*C$ effect need not be changed. The corresponding alterations would need to be made to the second contrast defined on lines 78 to 87. Finally, to fix the second-order effect at A3 in the third LMATRIX, the researcher would simply move the lines of syntax described above down two lines instead of one.


The output obtained from running the syntax in Figure B2 shows that the second-order simple effect of B at A1 and C1 is significant ($p < .001$) while the other two simple effects are not ($p > .05$). Also provided in the output are the results for the two one degree of freedom linearly independent contrasts, $L1_B$ and $L2_B$. These contrasts may be useful to the researcher, as they provide second-order simple comparisons that help explain the locus of the significant second-order simple effect.

Appendix C

Testing simple interaction and second-order simple effects in a three-way mixed-design ANOVA using LMATRIX and MMATRIX

Testing simple interaction effects in a three-way mixed-design ANOVA can be less complicated than its between-subjects counterpart, especially when the design has one between- and two within-subjects variables. The number of LMATRIX subcommands required is equal to the number of levels for the variable that is being held constant. Suppose a researcher has a $3 \times 3 \times (3)$ ($A[1,2,3] \times B[1,2,3] \times C[1,2,3]$) design, where C is a within-subjects variable, and wishes to test the simple $A \times C$ interaction at each level of B. In this case, the researcher would specify three LMATRIX


subcommands, one for each level of B. The number of contrasts the researcher must define within each LMATRIX is equivalent to the number of degrees of freedom for the between-subjects variable(s) involved in the simple interaction. Following our example, the researcher needs to specify two contrasts for each simple interaction effect (i.e., $df_A = (a - 1) = (3 - 1) = 2$). The number of MMATRIX subcommands required depends on the role that the within-subjects variable plays. Specifically, if it is involved in the simple interaction, as it is in the simple $A \times C$ interaction, only one MMATRIX is required. In this case, the number of contrasts defined within the MMATRIX is equal to the number of degrees of freedom for the within-subjects portion of the simple interaction (i.e., $df_C = (c - 1) = (3 - 1) =$



```

7  /LMATRIX 'simple interaction A x C at B1'
8      A 1 -1 0
9      A*B 1 0 0
10     -1 0 0
11     0 0 0;
12     A 1 0 -1
13     A*B 1 0 0
14     0 0 0
15     -1 0 0;
16  /LMATRIX 'simple interaction A x C at B2'
17     A 1 -1 0
18     A*B 0 1 0
19     0 -1 0
20     0 0 0;
21     A 1 0 -1
22     A*B 0 1 0
23     0 0 0
24     0 -1 0;
25  /LMATRIX 'simple interaction A x C at B3'
26     A 1 -1 0
27     A*B 0 0 1
28     0 0 -1
29     0 0 0;
30     A 1 0 -1
31     A*B 0 0 1
32     0 0 0
33     0 0 -1;
34  /MMATRIX 'pooled C differences'
35     all 1 -1 0;
36     all 1 0 -1.
  
```

Figure C1. LMATRIX and MMATRIX syntax for testing the simple $A \times C$ interactions at each level of B when the design is comprised of two between- and one within-subjects variable.



```

88  GLM W1 W2 W3 BY A B
89  /WVSFACTOR=C 3
90  /METHOD=SSTYPE(3)
91  /CRITERIA=ALPHA(.05)
92  /LMATRIX "simple interaction A x B"
93     A*B 1 -1 0
94     -1 1 0
95     0 0 0;
96     A*B 1 0 -1
97     -1 0 1
98     0 0 0;
99     A*B 1 -1 0
100    0 0 0
101    -1 1 0;
102    A*B 1 0 -1
103    0 0 0
104    -1 0 1;
105  /MMATRIX 'fix at C1'
106    all 1 0 0.
  
```

Figure C2. LMATRIX and MMATRIX syntax for testing the simple interaction $A \times B$ at C1 when the design is comprised of two between- and one within-subjects variables.

2). The simple interaction $A \times C$ is represented in the combination of the weighting coefficients provided in the LMATRIX and MMATRIX subcommands. Figure C1 displays the syntax for testing the simple $A \times C$ interactions.

As can be seen in Figure C1, each of the three LMATRIX subcommands is comprised of two linearly independent contrasts on A. More specifically, the L1_A contrast from Appendix A is defined on lines 8, 17, and 26, while contrast L2_A is defined on lines 12, 21, and 30. The reader may verify by comparing the A*B effects in each LMATRIX that the

only difference between each subcommand is that the two linearly independent contrasts on A are fixed at different levels of B, with the first LMATRIX being fixed at B1 (the contrast on A is represented in the first column), the second at B2 (represented in the second column), and the third at B3 (represented in the third column). The MMATRIX subcommand on lines 34 to 36 defines the two linearly independent contrasts on the within-subject variable C. The combination of the LMATRIX and MMATRIX generates separate tests of the simple interaction $A \times C$ at B1, B2, and B3. When the syntax is run, the test results for the respective simple interaction effects are located in the Multivariate Tests Results table in the Custom Hypothesis Tests #1, #2, and #3 sections.

For second-order simple effects on a between-subjects variable, the researcher must create as many LMATRIX subcommands as there are levels of the between-subjects variable that is being held constant. For illustrative purposes, we will explore the second-order simple effect of A at the different levels of B and C1. First, the researcher must create three LMATRIX subcommands, one for each level of B. Within each LMATRIX, the researcher must generate as many linearly independent contrasts as there are degrees of freedom for the second-order simple effect ($df_A = (3 - 1) = 2$). For each of these contrasts, the second-order simple effect must be defined on its own (i.e., A) and in higher-order order effects involving only between-subjects variables (i.e., A*B). The MMATRIX subcommand would simply fix the test at the prescribed level of the within-subjects variable. The syntax to run these tests is identical to Figure C1, except that the MMATRIX would fix the level of C where the effect is supposed to be tested (i.e., for second-order simple effect of A at B1 and C1, the MMATRIX would look like this "all 1 0 0.") and the labels in quotation marks should be changed to reflect their new meaning. If the syntax were run, the output would include three Custom Hypothesis Tests sections with the results of the respective simple interaction tests in the Tests Results tables.

In contrast to the preceding example, the reader may be interested in the case where the within-subjects variable is being held constant, instead of a between-subjects variable (e.g., testing the simple $A \times B$ interaction effect at each level of C). In this case, the number of MMATRIX subcommands

required to test simple interaction effects is the same as the number of levels of the within-subjects variable, with one subcommand to fix the test at each level of the within-subjects variable. Note, however, only one MMATRIX subcommand may be specified for each GLM command. If more than one is provided, only the last one listed in the syntax would be used in computations. Therefore, the researcher is required to create three separate GLM commands (a simple copy and paste works well). For the LMATRIX portion of the syntax, there needs to be one LMATRIX with as many linearly independent contrasts as there are degrees of freedom for the simple A*B interaction (i.e., $df_{AB} = 4$). The set of four linearly independent contrasts derived in Appendix A may be used (see Figure A1). The reader may find in Figure C2 the syntax used to test the simple interaction $A \times B$ at C1. In order to test the simple interaction $A \times B$ at C2 and C3, the reader would simply copy and paste the syntax in Figure C2 then change the MMATRIX to "all 0 1 0." for a test at C2 and paste again and change the MMATRIX to "all 0 0 1." for a test at C3.

For follow-up second-order simple effects tests, the researcher may modify the syntax provided in Figure C1. For instance, the level at which C is being held constant may be changed easily by altering the MMATRIX statement on lines 35 to 36. If the second-order simple effect of B at the different levels of A and C1 were of interest, the MMATRIX statement would be changed to "all 1 0 0." in order to fix C at level 1. Second, all contrasts on A (i.e., lines 8, 12, 17, 21, etc) would need to be changed to B. Third, the A*B effects (i.e., 9, 13, 18, 22, etc) would need to be changed to reflect that the contrasts on B are fixed at levels of A, instead of the contrasts on A being fixed at levels of B, as they are in Figure C1. Before making the alteration, the reader may recall that the rows represent the different levels of A and the columns represent the different levels of B. To make the change of interest for the first A*B effect on lines 9 to 11, the contrast on B (i.e., 1 -1 0) must appear at the first row of the matrix (i.e., the row representing A1) and all other values must be set to zero. It is advised that the reader refer to the matrices displayed in Figure A1, or those generated for the reader's particular circumstance, before making such alterations.

Appendix D

Testing simple interaction effects in a three-way within-subjects ANOVA using MMATRIX

For any three-way within-subjects ANOVA, testing the simple interaction effects requires as many MMATRIX subcommands as there are levels for the within-subjects variable that is being held constant. Suppose a researcher has a $(3) \times (3) \times (3)$ ($A[1,2,3] \times B[1,2,3] \times C[1,2,3]$) within-subjects design and wishes to test the simple interaction $B \times C$ at each level of A. In this case, the researcher requires three MMATRIX subcommands, one for each level of A, the variable that is being held constant. Note that SPSS only computes one MMATRIX for each GLM. Thus, each MMATRIX subcommand must be defined in a separate GLM. The number of contrasts the researcher must define in each MMATRIX is equal to the number of degrees of freedom for the simple interaction effect being tested. In this case, the researcher needs four linearly independent contrasts ($df_{BC} = (b - 1) \times (c - 1) = 4$) defined using weighting coefficients. The set of contrasts and weighting coefficients put forward in Figure A1 may be used, but careful attention must be spent when translating them into the MMATRIX subcommands.

Figure D1 displays the syntax for testing the simple interaction $B \times C$ at A1. The researcher may first note that

there is one MMATRIX with four contrasts. The first contrast ranges from line 18 to 20, the second from line 21 to 23, etc. The weighting coefficients are organized in a matrix format with a specific configuration unique to the order of the dependent variables (i.e., levels of the within-subjects variables) listed in the line of syntax that defines the GLM (see lines 11 and 12). Note that this order was generated when the within-subjects variables were defined in the “Repeated Measures Define Factor(s)” dialog box before the syntax was pasted. Specifically, the rows represent the levels of the first defined variable (i.e., A1 on line 18, A2 on line 19, and A3 on line 20) and the columns represent combinations of the second and third defined variables (i.e., B and C), with the second variable changing slower than the third in the left to right order (i.e., B1C1, B1C2, B1C3, B2C1 ... B3C3). The organization of the variables in this manner is only to facilitate writing, comparing, and altering contrasts. The researcher could instead use one line of syntax with weighting coefficients that correspond to the specific left to right ordering of the levels of the within-subjects variables listed in the GLM, but may quickly discover the difficulty with writing, interpreting, and altering the contrasts that way.

With an understanding of how the weighting coefficients are organized, the reader may see that the four contrasts in Figure D1 all have 0s for the rows that

```

11 GLM A1B1C1 A1B1C2 A1B1C3 A1B2C1 A1B2C2 A1B2C3 A1B3C1 A1B3C2 A1B3C3 A2B1C1 A2B1C2 A2B1C3 A2B2C1
12 A2B2C2 A2B2C3 A2B3C1 A2B3C2 A2B3C3 A3B1C1 A3B1C2 A3B1C3 A3B2C1 A3B2C2 A3B2C3 A3B3C1 A3B3C2 A3B3C3
13 /WVSFACTOR=A 3 Polynomial B 3 Polynomial C 3 Polynomial
14 /METHOD=SSTYPE(3)
15 /PLOT=PROFILE(A*B*C)
16 /CRITERIA=ALPHA(.05)
17 /MMATRIX 'simple interaction B x C at A1'
18 all 1 -1 0 -1 1 0 0 0 0
19 0 0 0 0 0 0 0 0 0
20 0 0 0 0 0 0 0 0 0;
21 all 1 -1 0 0 0 0 -1 1 0
22 0 0 0 0 0 0 0 0 0
23 0 0 0 0 0 0 0 0 0;
24 all 1 0 -1 -1 0 1 0 0 0
25 0 0 0 0 0 0 0 0 0
26 0 0 0 0 0 0 0 0 0;
27 all 1 0 -1 0 0 0 -1 0 1
28 0 0 0 0 0 0 0 0 0
29 0 0 0 0 0 0 0 0 0.

```

Figure D1. MMATRIX for testing the simple interaction $B \times C$ when the design is completely within-subjects.

are associated with A2 and A3, as they should be given that the syntax tests the simple interaction $B \times C$ at A1. Looking at the first contrast on lines 18 to 20, the reader may note that the weighting coefficients for the A1 row (i.e., line 18) correspond to the contrast defined in the top left table in Figure A1. It is also the case that the second, third, and fourth contrasts correspond to the weighting coefficients in the bottom left, top right, and bottom right tables respectively in Figure A1. For testing the simple interaction $B \times C$ at A2, the researcher may swap the weighting coefficients on lines 18, 21, 24, and 27 with those from one line below (i.e., lines that correspond to A2), in order to fix the contrasts at A2. Likewise, to test the simple interaction $B \times C$ at A3, the researcher may swap the coefficients from two lines below (i.e., lines that correspond to A3). When the syntax is run, the output contains a Custom Hypothesis Tests section and the results for the simple interaction effects test are in the Multivariate Test Results table.

Testing second-order simple effects in a three-way within-subjects ANOVA using MMATRIX

The number of MMATRIX subcommands required to test second-order simple effects is equivalent to the number of levels where the effects are being tested. Suppose that a researcher wishes to follow-up on a significant simple interaction $B \times C$ at A1 with a set of simple effects tests of C at the different levels of B and A1. Here, the researcher would need three separate MMATRIX subcommands, with each one defined in a different GLM. As seen previously, the number of

linearly independent contrasts needed within each MMATRIX is equal to the number of degrees of freedom for the second-order simple effect. With two degrees of freedom for second-order simple effect of C, the researcher requires two linearly independent contrasts.

Figure D2 displays the MMATRIX syntax for testing the second-order simple effect of C at B1 and A1. It consists of two linearly independent contrasts with weighting coefficients laid out as described in the previous section. The first contrast on line 90 to 92 compares B1C1 with B1C2 at A1 while the second on line 93 to 95 compares B1C1 with B1C3 at A1. Together, these contrasts comprise the second-order simple effect of C at B1 and A1. To test the second-order effect of C at B2 and A1, the researcher can simply shift the 1 and -1 for both contrasts to the right three digits. Likewise, for the second-order simple effect of C at B3 and A1, the researcher can shift the 1 and -1 to the right six digits. In the SPSS output, there are three sets of results, one for each GLM. Within each set, a Custom Hypothesis Tests section that contains a Contrast Results and a Multivariate Test Results table can be found. In the Multivariate Test Results table, the researcher can find the second-order simple effect test results. In the Contrast Results table, the researcher can find the results for the two contrasts defined in the MMATRIX subcommand. These are actually the second-order simple comparisons that can be examined following a significant second-order simple effect test.

```

83 GLM A1B1C1 A1B1C2 A1B1C3 A1B2C1 A1B2C2 A1B2C3 A1B3C1 A1B3C2 A1B3C3 A2B1C1 A2B1C2 A2B1C3 A2B2C1
84 A2B2C2 A2B2C3 A2B3C1 A2B3C2 A2B3C3 A3B1C1 A3B1C2 A3B1C3 A3B2C1 A3B2C2 A3B2C3 A3B3C1 A3B3C2 A3B3C3
85 /WSFACTOR=A 3 Polynomial B 3 Polynomial C 3 Polynomial
86 /METHOD=SSTYPE(3)
87 /PLOT=PROFILE(A*B*C)
88 /CRITERIA=ALPHA(.05)
89 /MMATRIX 'second-order simple effect of C at B1 and A1'
90     all 1 -1 0 0 0 0 0 0
91         0 0 0 0 0 0 0 0
92         0 0 0 0 0 0 0 0;
93     all 1 0 -1 0 0 0 0 0
94         0 0 0 0 0 0 0 0
95         0 0 0 0 0 0 0 0.

```

Figure D2. MMATRIX for testing the second-order simple effect of C at B1 and A1 when the design is fully within-subjects.